# Taylor Series Model

In[144]:=
```
S[x_] = Normal[Series[f[x], {x, c, 4}]] (*Taylor Series expansion*)
```

Out[144]= $f[c] + (-c+x)\, f'[c] + \frac{1}{2}(-c+x)^2 f''[c] + \frac{1}{6}(-c+x)^3 f^{(3)}[c] + \frac{1}{24}(-c+x)^4 f^{(4)}[c]$

In[145]:= $S\left[\frac{a+b}{2}\right]$ `(*Example*)`

Out[145]= $f[c] + \left(\frac{a+b}{2}-c\right) f'[c] + \frac{1}{2}\left(\frac{a+b}{2}-c\right)^2 f''[c] + \frac{1}{6}\left(\frac{a+b}{2}-c\right)^3 f^{(3)}[c] + \frac{1}{24}\left(\frac{a+b}{2}-c\right)^4 f^{(4)}[c]$

# Definite Integral of Taylor Series

In[146]:=
```
Exact = Simplify[Integrate[S[x], {x, a, b}]] (*Exact integral of S(x) on [a,b]*)
```

Out[146]=
$$-a\, f[c] + b\, f[c] - \frac{1}{2} a^2 f'[c] + \frac{1}{2} b^2 f'[c] + a\, c\, f'[c] - b\, c\, f'[c] - \frac{1}{6}(a-b)\left(a^2+b^2+a\,(b-3c) - 3\,b\,c + 3\,c^2\right) f''[c] + \frac{1}{24}\left(-(a-c)^4 + (b-c)^4\right) f^{(3)}[c] + \frac{1}{120}\left((b-c)^5 + (-a+c)^5\right) f^{(4)}[c]$$

# Midpoint Integration

In[148]:= `Midpoint = Simplify`$\left[(b-a)\, S\left[\frac{a+b}{2}\right]\right]$

Out[148]=
$$(-a+b)\left(f[c] + \frac{1}{2}(a+b-2c)\, f'[c] + \frac{1}{8}(a+b-2c)^2 f''[c] + \frac{1}{48}(a+b-2c)^3 f^{(3)}[c] + \frac{1}{384}(a+b-2c)^4 f^{(4)}[c]\right)$$

In[152]:=
```
Simplify[Midpoint - Exact]
```

Out[152]=
$$\frac{1}{1920}(a-b)^3 \left(80\, f''[c] + 40\,(a+b-2c)\, f^{(3)}[c] + \left(11\,a^2 + 18\,a\,b + 11\,b^2 - 40\,a\,c - 40\,b\,c + 40\,c^2\right) f^{(4)}[c]\right)$$

In[156]:=
```
80 / 1920
```

Out[156]= $\frac{1}{24}$

# Trapezoid Integration

In[153]:=
```
Trap = Simplify[(b - a) (S[a] + S[b])/2] (*Series expansion of trapezoid rule about c*)
```

Out[153]=
$$\frac{1}{48}(-a+b)\left(48 f[c]+24(a-c) f'[c]+24(b-c) f'[c]+12(a-c)^2 f''[c]+12(b-c)^2 f''[c]+4(a-c)^3 f^{(3)}[c]+4(b-c)^3 f^{(3)}[c]+(a-c)^4 f^{(4)}[c]+(b-c)^4 f^{(4)}[c]\right)$$

In[155]:=
```
Simplify[Trap - Exact]
```

Out[155]=
$$-\frac{1}{240}(a-b)^3\left(20 f''[c]+10(a+b-2c) f^{(3)}[c]+\left(3a^2+4ab+3b^2-10ac-10bc+10c^2\right) f^{(4)}[c]\right)$$

```
(*How does this compare to Midpoint??*)
```

# Simpson Rule

In[158]:=
```
Simp = Simplify[(b - a)/6 (S[a] + 4 S[(a + b)/2] + S[b])]
```

Out[158]=
$$\frac{1}{144}(-a+b)\left(48 f[c]+24(a-c) f'[c]+24(b-c) f'[c]+12(a-c)^2 f''[c]+12(b-c)^2 f''[c]+4(a-c)^3 f^{(3)}[c]+4(b-c)^3 f^{(3)}[c]+(a-c)^4 f^{(4)}[c]+(b-c)^4 f^{(4)}[c]+96\left(f[c]+\frac{1}{2}(a+b-2c) f'[c]+\frac{1}{8}(a+b-2c)^2 f''[c]+\frac{1}{48}(a+b-2c)^3 f^{(3)}[c]+\frac{1}{384}(a+b-2c)^4 f^{(4)}[c]\right)\right)$$

In[160]:=
```
Simplify[Simp - Exact]
```

Out[160]=
$$-\frac{(a-b)^5 f^{(4)}[c]}{2880}$$

# Finite Differences

In[161]:=
```
FwdDiff = Simplify[(S[c + h] - S[c])/h]
```

Out[161]=
$$f'[c]+\frac{1}{24} h\left(12 f''[c]+h\left(4 f^{(3)}[c]+h f^{(4)}[c]\right)\right)$$

In[162]:=
```
CenteredDiff = Simplify[(S[c + h] - S[c - h])/(2 h)]
```

Out[162]=
$$f'[c]+\frac{1}{6} h^2 f^{(3)}[c]$$

In[163]:= `CenteredDiff2nd = Simplify[(S[c + h] - 2 S[c] + S[c - h])/h^2]`

Out[163]= $f''[c] + \frac{1}{12} h^2 f^{(4)}[c]$

# Richardson Extrapolation

In[164]:= `S[x_] = Normal[Series[f[x], {x, c, 14}]] (*Longer Taylor Series expansion*)`

Out[164]=
$$f[c] + (-c + x) f'[c] + \frac{1}{2} (-c + x)^2 f''[c] + \frac{1}{6} (-c + x)^3 f^{(3)}[c] + \frac{1}{24} (-c + x)^4 f^{(4)}[c] + \frac{1}{120} (-c + x)^5 f^{(5)}[c] + \frac{1}{720} (-c + x)^6 f^{(6)}[c] + \frac{(-c + x)^7 f^{(7)}[c]}{5040} + \frac{(-c + x)^8 f^{(8)}[c]}{40\,320} + \frac{(-c + x)^9 f^{(9)}[c]}{362\,880} + \frac{(-c + x)^{10} f^{(10)}[c]}{3\,628\,800} + \frac{(-c + x)^{11} f^{(11)}[c]}{39\,916\,800} + \frac{(-c + x)^{12} f^{(12)}[c]}{479\,001\,600} + \frac{(-c + x)^{13} f^{(13)}[c]}{6\,227\,020\,800} + \frac{(-c + x)^{14} f^{(14)}[c]}{87\,178\,291\,200}$$

In[165]:= `DS[h_] = Simplify[(S[c + h] - S[c - h])/(2 h)] (*Centered Difference Approx.*)`

Out[165]=
$$f'[c] + \frac{1}{6} h^2 f^{(3)}[c] + \frac{1}{6\,227\,020\,800} h^4 \left(51\,891\,840 f^{(5)}[c] + 1\,235\,520 h^2 f^{(7)}[c] + 17\,160 h^4 f^{(9)}[c] + 156 h^6 f^{(11)}[c] + h^8 f^{(13)}[c]\right)$$

In[166]:= `Expand[DS[h]]`

Out[166]=
$$f'[c] + \frac{1}{6} h^2 f^{(3)}[c] + \frac{1}{120} h^4 f^{(5)}[c] + \frac{h^6 f^{(7)}[c]}{5040} + \frac{h^8 f^{(9)}[c]}{362\,880} + \frac{h^{10} f^{(11)}[c]}{39\,916\,800} + \frac{h^{12} f^{(13)}[c]}{6\,227\,020\,800}$$

In[167]:= `Expand[3^2 DS[h/3]]`

Out[167]=
$$9 f'[c] + \frac{1}{6} h^2 f^{(3)}[c] + \frac{h^4 f^{(5)}[c]}{1080} + \frac{h^6 f^{(7)}[c]}{408\,240} + \frac{h^8 f^{(9)}[c]}{264\,539\,520} + \frac{h^{10} f^{(11)}[c]}{261\,894\,124\,800} + \frac{h^{12} f^{(13)}[c]}{367\,699\,351\,219\,200}$$

In[168]:= `DS2[h_] = Expand[(DS[h] - 3^2 DS[h/3])/(1 - 3^2)]`

Out[168]=
$$f'[c] - \frac{h^4 f^{(5)}[c]}{1080} - \frac{h^6 f^{(7)}[c]}{40\,824} - \frac{13 h^8 f^{(9)}[c]}{37\,791\,360} - \frac{41 h^{10} f^{(11)}[c]}{13\,094\,706\,240} - \frac{671 h^{12} f^{(13)}[c]}{33\,427\,213\,747\,200}$$

## Continuing Recursively...

In[169]:= 
$$\text{DS4}[h\_] = \text{Expand}\left[\frac{\text{DS2}[h] - 3^4\,\text{DS2}\left[\frac{h}{3}\right]}{1 - 3^4}\right]$$

Out[169]= 
$$f'[c] + \frac{h^6 f^{(7)}[c]}{3\,674\,160} + \frac{13\,h^8 f^{(9)}[c]}{3\,061\,100\,160} + \frac{533\,h^{10} f^{(11)}[c]}{13\,637\,201\,212\,800} + \frac{27\,511\,h^{12} f^{(13)}[c]}{109\,657\,974\,697\,689\,600}$$

In[170]:= 
$$\text{DS6}[h\_] = \text{Expand}\left[\frac{\text{DS4}[h] - 3^6\,\text{DS4}\left[\frac{h}{3}\right]}{1 - 3^6}\right]$$

Out[170]= 
$$f'[c] - \frac{h^8 f^{(9)}[c]}{192\,849\,310\,080} - \frac{41\,h^{10} f^{(11)}[c]}{773\,229\,308\,765\,760} - \frac{27\,511\,h^{12} f^{(13)}[c]}{79\,940\,663\,554\,615\,718\,400}$$

In[171]:= 
$$\text{DS8}[h\_] = \text{Expand}\left[\frac{\text{DS6}[h] - 3^8\,\text{DS6}\left[\frac{h}{3}\right]}{1 - 3^8}\right]$$

Out[171]= 
$$f'[c] + \frac{h^{10} f^{(11)}[c]}{139\,181\,275\,577\,836\,800} + \frac{671\,h^{12} f^{(13)}[c]}{12\,950\,387\,495\,847\,746\,380\,800}$$

In[172]:= 
$$\text{DS10}[h\_] = \text{Simplify}\left[\frac{\text{DS8}[h] - 3^{10}\,\text{DS8}\left[\frac{h}{3}\right]}{1 - 3^{10}}\right] \quad \text{(* An } O\left(h^{12}\right) \text{ finite difference approx! *)}$$

Out[172]= 
$$f'[c] - \frac{h^{12} f^{(13)}[c]}{1\,282\,088\,362\,088\,926\,891\,699\,200}$$

## Romberg Integration

In[173]:= 
$$g[x\_] := 4\,\text{Sqrt}\left[1 - x^2\right]$$

In[174]:= 
```
Integrate[g[x], {x, 0, 1}]
```

Out[174]= $\pi$

```
In[178]:= NIntegrate[g[x], {x, 0, 1}, WorkingPrecision → 1000]
```

NIntegrate: NIntegrate failed to converge to prescribed accuracy after 9 recursive bisections in x near {x} = {≪1058≫}. NIntegrate obtained ≪1057≫ and ≪1063≫ for the integral and error estimates.

```
Out[178]= 3.1415926535897932384626433832795028841971693993751058209749445923078164062862089⋱
 98628034825342117067982148086513282306647093844609550582231725359408128481117450⋱
 28410270193852110555964462294895493038196442881097566593344612847564823378678316⋱
 52712019091456485669234603486104543266482133936072602491412737245870066063155881⋱
 74881520920962829254091715364367892590360011330530548820466521384146951941511609⋱
 43305727036575959195309218611738193261179310511854807446237996274956735188575272⋱
 48912279381830119491298336733624406566430860213949463952247371907021798609437027⋱
 70539217176293176752384674818467669405132000568127145263560827785771342757789609⋱
 17363717872146844090122495343014654958537105079227968925892354201995611212902196⋱
 08640344181598136297747713099605187072113499999983729780499510597317328160963185⋱
 95024459455346908302642522308253344685035261931188171010003137838752886587533208⋱
 38142061717766914730359825349042875546873115956286388235378759375195778185778053⋱
 2171226806613001927876611195909216420199
```